

# 17

## *Redistributive Effect of Health Finance*

Contributions toward the finance of health care may redistribute disposable income. This redistribution may be intended or unintended. Even in the latter case, policy makers may be interested in the degree to which it occurs because of consequences for the distribution of goods and services other than health care and, ultimately, for welfare. Redistribution can occur when payments toward the financing of health care are compulsory and independent of utilization, most obviously when health care is partly financed from government tax revenues. If tax liabilities rise disproportionately with gross incomes, then the posttax distribution of income will be more equal than the pretax distribution. When health care payments are made voluntarily, they do not have a redistributive effect on economic welfare. Payments are made directly in return for a product—health care. It would not make sense to consider the welfare-reducing effect of the payments made while ignoring the welfare-increasing effect of the health care consumption deriving from those payments. This begs the question of the extent to which out-of-pocket payments for health care should be considered voluntary. It might be argued that the moral compulsion to purchase vital health care for a relative is no less strong than the legal compulsion to pay taxes. But in most instances, there is discretion in the purchase of health care in response to health problems.

Redistribution can be vertical and horizontal. The former occurs when payments are disproportionately related to ability to pay. The extent of vertical redistribution can be inferred from measures of progressivity discussed in the previous chapter. Horizontal redistribution occurs when persons with equal ability to pay contribute unequally to health care payments. In this chapter, we describe how the total redistributive effect of compulsory health payments can be measured and how this redistribution can be decomposed into its vertical and horizontal components.

### **Decomposing the redistributive effect**

One way of measuring the redistributive effect of any compulsory payment on the distribution of incomes is to compare inequality in prepayment incomes—as measured by, for instance, the Gini coefficient—with inequality in postpayment incomes (Lambert 1989). The redistributive impact can be defined as the reduction in the Gini coefficient caused by the payment. Thus,

$$(17.1) \quad RE = G^X - G^{X-P},$$

where  $G^X$  and  $G^{X-P}$  are the prepayment and postpayment Gini coefficients, respectively, where  $X$  denotes prepayment income, or more generally some measure of

ability to pay, and  $P$  denotes the payment. Aronson, Johnson, and Lambert (1994) have shown that this difference can be written as

$$(17.2) \quad RE = V - H - R,$$

where  $V$  is vertical redistribution,  $H$  is horizontal inequity, and  $R$  is the degree of reranking. Because there are few households in any sample with exactly the same prepayment income, one needs to artificially create groups of prepayment equals, within intervals of prepayment income, to distinguish and compute the components of equation 17.2. The vertical redistribution component, which represents the redistribution that would arise if there were horizontal equity in payments, can then be defined as

$$(17.3) \quad V = G^X - G^0,$$

where  $G^0$  is the between-groups Gini coefficient for postpayment income. This can be computed by replacing all postpayment incomes with their group means.  $V$  itself can be decomposed into a payment rate effect and a progressivity effect,

$$(17.4) \quad V = \left( \frac{g}{1-g} \right) K_E,$$

where  $g$  is the sample average payment rate (as a proportion of income) and  $K_E$  is the Kakwani index of payments that would arise if there were horizontal equity in health care payments. It is computed as the difference between the between-groups concentration index for payments and  $G^X$ . In effect, the vertical redistribution generated by a given level of progressivity is "scaled" by the average rate  $g$ .

Horizontal inequity  $H$  is measured by the weighted sum of the group ( $j$ ) specific postpayment Gini coefficients,  $G_j^{X-P}$ , where weights are given by the product of the group's population share and its postpayment income share,  $\alpha_j$ .

$$(17.5) \quad H = \sum_j \alpha_j G_j^{X-P}.$$

Note that because the Gini coefficient for each group of prepayment equals is nonnegative,  $H$  is also nonnegative. Because it is subtracted in equation 17.2, horizontal inequity  $H$  can only reduce redistribution, not increase it. This simply implies that any horizontal inequity will always make a postpayment distribution of incomes more unequal than it would have been in its absence.

Finally,  $R$  captures the extent of reranking of households that occurs in the move from the prepayment to the postpayment distribution of income. It is measured by

$$(17.6) \quad R = G^{X-P} - C^{X-P},$$

where  $C^{X-P}$  is a postpayment income concentration index that is obtained by first ranking households by their prepayment incomes and then, within each group of prepayment "equals," by their postpayment income. Note again that  $R$  cannot be negative, because the concentration curve of postpayment income cannot lie below the Lorenz curve of postpayment income. The two curves coincide (and the two indices are equal) if no reranking occurs.

All in all, the total redistributive effect can be decomposed into four components: an average rate effect ( $g$ ), the departure-from-proportionality or progressivity effect ( $K_E$ ), a horizontal inequity effect  $H$ , and a reranking effect  $R$ . Practical execution of this decomposition requires an arbitrary choice of income intervals to define "equals." Although this choice will not affect the total  $H+R$ , it will affect the relative magnitudes of  $H$  and  $R$ . In general, the larger are the income intervals, the greater will be the estimate of horizontal inequity and the smaller will be the

estimate of reranking (Aronson, Johnson, and Lambert 1994). That makes the distinction between  $H$  and  $R$  rather uninteresting in applications.<sup>1</sup> More interesting is the quantification of the vertical redistribution  $V$ , both in absolute magnitude and relative to the total redistributive effect, and its separation into the average rate and progressivity effects. Van Doorslaer et al. (1999) make this decomposition of the redistributive effect of health finance for 12 OECD countries.

**Box 17.1** *Redistributive Effect of Public Finance of Health Care in the Netherlands, the United Kingdom, and the United States*

To illustrate the redistributive effect of health finance and its decomposition, we present results for three countries—the Netherlands, the United Kingdom, and the United States—taken from van Doorslaer et al. (1999). For each country, we show the redistributive effect of compulsory payments toward publicly financed health care. Public finance predominates in the finance of health care in both the Netherlands and the United Kingdom, but the source differs. The Netherlands relies mainly on social insurance, whereas most finance in the United Kingdom comes from general taxation. Although the majority of health care finance is private in the United States, there is a substantial contribution from public funds, with two-thirds of this from general taxation.

The figures in the first row of the table indicate that public finance of health care brings about redistribution from rich to poor in the United Kingdom and the United States but from poor to rich in the Netherlands. In both the United Kingdom and the United States, vertical redistribution is very large in comparison with the total redistribution. If there were no horizontal inequity, redistribution from rich to poor would be only 2.4 percent and 5 percent greater than its actual magnitude in the United Kingdom and United States, respectively. In the Netherlands, vertical redistribution is from poor to rich, and horizontal inequity and reranking adds a further 6.6 percent of the redistribution in that direction. In absolute value, the redistribution is largest in the Netherlands because public payments for health care are larger relative to income—8.2 percent of income, compared with only 3.6 percent in the United Kingdom and 6 percent in the United States. It is interesting that the United States spends relatively more public dollars on health care than does the United Kingdom, despite the United Kingdom being a predominantly publicly funded system. This difference in the scale of public spending is responsible for the greater redistributive effect in the United States. Public finance is more progressive in the United Kingdom, indicated by the Kakwani index, but there is less of it.

Source: Authors.

*Decomposition of Redistributive Effect of Public Finance of Health Care in the Netherlands, the United Kingdom, and the United States*

		Netherlands (1992)	United Kingdom (1992)	United States (1987)
Redistributive effect	$RE = GX - GX - P$	-0.0096	0.0044	0.0063
Vertical redistribution effect	$V = [g/(1-g)] * KE$	-0.0089	0.0045	0.0066
Vertical redistribution as % of RE	$(V/RE) * 100$	93.40	102.40	105.00
Total payment as fraction of income	$g$	0.0821	0.0361	0.0604
Kakwani index assuming horizontal equity	$KE$	-0.0999	0.1221	0.0979

Source: van Doorslaer et al. 1999.

<sup>1</sup>See Duclos, Jalbert, and Araar (2003) for an alternative approach that avoids this limitation.

## Computation

Let  $y$  be prepayment income and  $wt$  be the sample weight variable. Create the weighted fractional rank ( $r$ ), and estimate the Gini coefficient ( $gini$ ) for prepayment income using, for example, the covariance approach (see chapter 8),

```
egen rank1 = rank(y), unique
sort rank1
qui sum wt
gen wi=wt/r(sum)
gen cusum=sum(wi)
gen wj=cusum[_n-1]
replace wj=0 if wj==.
gen r=wj+0.5*wi

qui sum y [aw=wt]
sca m_y=r(mean)
qui cor r y [aw=wt], c
sca gini=2*r(cov_12)/m_y
```

Let  $X$  be a global containing all the compulsory health payments variables for which the decomposition is to be undertaken. For taxes, we wish to identify the redistributive effect only of that part of taxation that is used to fund health care. So, all tax payments must be scaled by tax-funded expenditure on health care as a proportion of aggregate general government expenditure on all goods and services. Generate a variable representing postpayment income for each payment, and estimate the Gini coefficient for that variable. Finally, compute the redistributive effect for each payment as the difference between the pre- and postpayment Gini indices. This can all be done in the following loop:

```
foreach x of global X {
  qui {
    gen ypost_`x`=y-`x'
    sum ypost_`x' [aw=wt]
    sca my_`x`=r(mean)
    egen rank_`x' = rank(ypost_`x'), unique
    sort rank_`x'
    gen cusum_`x`=sum(wi)
    gen wj_`x`=cusum_`x'[_n-1]
    replace wj_`x`=0 if wj_`x'==.
    gen r_`x`=wj_`x'+0.5*wi
    corr r_`x' ypost_`x' [aw=wt], c
    sca gini_`x`=2*r(cov_12)/my_`x'
    sca re_`x`=gini-gini_`x'
  }
}
```

For the decomposition of the redistributive effect, households must be grouped into prepayment “equals.” To do this, create a variable that categorizes households according to prepayment income intervals of fixed width. For example, to break the sample into 100 groups, each spanning an interval of income of fixed width, the following may be used:

```

qui sum y
local max=r(max)
kdensity y [aw=wt], n(100) nograph
local width=r(scale)
egen ygroup=cut(y), at(0(`width')`max') icodes
recode ygroup .=99

```

where the `kdensity` command is used simply to create the width of the income intervals and the `egen` command creates the categorical variable, `ygroup`.

To compute the concentration index of postpayment income, which is subtracted from the Gini coefficient for prepayment income in calculation of the reranking term (equation 17.6), we need to rank the groups by prepayment income and then rank households within the groups by postpayment income. With households ranked in this way, the appropriate weighted fractional rank must be computed. The concentration index can then be estimated by the covariance method and the reranking term computed. This is all done in the following loop:

```

foreach x of global X {
    qui {
        drop cusum_`x' wj_`x' r_`x'
        sort ygroup rank_`x'
        gen cusum_`x'=sum(wi)
        gen wj_`x'=cusum_`x'[_n-1]
        replace wj_`x'=0 if wj_`x'==.
        gen r_`x'=wj_`x'+0.5*wi
        corr r_`x' ypost_`x' [aw=wt], c
        sca ci_`x'=2*r(cov_12)/my_`x'
        sca rr_`x'=gini_`x' - ci_`x'
    }
}

```

To compute the Kakwani index in equation 17.4, the data can be collapsed to (weighted) group means and the between-groups concentration index for payments estimated at that level. First, create a constant (`grpsize`) that will indicate the group sizes when the data are collapsed, and `preserve` before collapsing the data so that they can be restored later to the household level.

```

gen grpsize=1
preserve
collapse (mean) y $X (sum) grpsize [aw=wt], by(ygroup)

```

At this level, the group sizes are the appropriate weights for computations at the level of group means. For these weights, create the weighted fractional rank to be used in estimation of the concentration index.

```

egen rank1 = rank(y), unique
sort rank1
qui sum grpsize
gen wi=grpsize/r(sum)
gen cusum=sum(wi)
gen wj=cusum[_n-1]
replace wj=0 if wj==.
gen r=wj+0.5*wi

```

Now the between-groups concentration index can be estimated and the Kakwani index computed as the difference between this and the Gini coefficient for prepayment income.

```
foreach x of global X {
    qui {
        sum `x' [aw=grpsize]
        sca m_`x'=r(mean)
        corr r `x' [aw=grpsize], c
        sca ci2_`x'=2*r(cov_12)/m_`x'
        sca k_`x'=ci2_`x' - gini
    }
}
```

The household-level data can then be restored with the `restore` command. The vertical redistribution effect (equation 17.4) can now be computed and this expressed as a percentage of the total redistribution effect.

```
foreach x of global X {
    qui sum `x' [aw=wt]
    sca g_`x'=r(mean)/m_y
    sca v_`x'=(g_`x'/(1-g_`x'))*k_`x'
    sca v100_`x'=(v_`x'/re_`x')*100
}
```

The results of the decomposition can then be displayed.

```
foreach x of global X {
    di "Decomposition of redistributive effect of `x'
    payments"
    di "Redistributive effect:", re_`x'
    di "Vertical redistribution:", v_`x'
    di "Vertical redistribution as % total redist. effect",
    v100_`x'
    di "Payments as a fraction of total income, g", g_`x'
    di "Horizontal inequity", v_`x'-rr_`x'-re_`x'
    di "Reranking", rr_`x'
}
```

## References

- Aronson, J. R., P. Johnson, and P. J. Lambert. 1994. "Redistributive Effect and Unequal Tax Treatment." *Economic Journal* 104: 262–70.
- Duclos, J.-Y., V. Jalbert, and A. Araar. 2003. "Classical Horizontal Inequity and Reranking: An Integrated Approach." In *Research on Economic Inequality: Fiscal Policy, Inequality and Welfare*, ed. J. A. Bishop and Y. Amiel, 65–100. Amsterdam, Netherlands: Elsevier.
- Lambert, P. J. 1989. *The Distribution and Redistribution of Income: A Mathematical Analysis*. Cambridge, MA: Blackwell.
- van Doorslaer, E., A. Wagstaff, H. van der Burg, T. Christiansen, G. Citoni, R. Di Biase, U. G. Gerdtham, M. Gerfin, L. Gross, U. Hakinnen, J. John, P. Johnson, J. Klavus, C. Lachaud, J. Lauritsen, R. Leu, B. Nolan, J. Pereira, C. Propper, F. Puffer, L. Rochaix, M. Schellhorn, G. Sundberg, and O. Winkelhake. 1999. "The Redistributive Effect of Health Care Finance in Twelve OECD Countries." *Journal of Health Economics* 18(3): 291–313.